3.1.  TIPOS DE DATOS SIMPLES PREDEFINIDOS
Los datos que podemos manejar en Turbo Pascal pueden ser de los siguientes tipos:

Tipos Simples


Tipo STRING

                                      Tipos de Datos             
Tipos Compuesto


Tipo Puntero


Tipos Procedural y Objeto


Veremos en este tema únicamente los tipos Simples, los cuales consisten en conjuntos de valores elementales; es decir, que no pueden dividirse, ni tratarse parcialmente, etc.. Un dato simple es, por ejemplo, un número entero, un número real, un carácter, etc..

Podemos clasificar los tipos simples de la siguiente manera:



                                                                                                       
Tipos Enteros

                                                                     Predefinidos              
El Tipo CHAR

                                              Tipos                                                
El Tipo BOOLEAN

          Tipos de datos          Ordinales

                                                                     Definibles por           
Tipos Abstractos

              Simples                                           el usuario                
Tipos Subrango


                                           Tipos Reales

      

Los tipos Ordinales están constituidos por valores elementales ordenados, cada uno de los cuales tiene asociado un ordinal (número entero que permite establecer una ordenación de estos valores).

      Los tipos Reales son ciertos subconjuntos del conjunto de los números reales, que veremos al final del tema.

      Los tipos Ordinales, a su vez, se dividen en Tipos Ordinales Predefinidos o conocidos y defi-nidos de antemano por el lenguaje, y los tipos que deben ser definidos por el usuario, si es que éste quiere hacer uso de un tipo de dato no estándar.

3.1.1.  Tipos de dato Enteros
Existen en Turbo Pascal cinco tipos numéricos enteros cuyos identificadores predefinidos son: SHORTINT, INTEGER, LONGINT, BYTE y WORD. Son los tipos utilizados para ma-nejar los datos numéricos enteros. Cada tipo abarca un rango de valores, de acuerdo con la si-guiente tabla:

TIPO
RANGO
TAMAÑO (bytes)

  SHORTINT

  INTEGER

  LONGINT

  BYTE

  WORD
-128 ... 127

-32.768 ... 32.767

-2.147.483.648 ... 2.147.483.647

                        0 ... 255

                        0 ... 65.535
1

2

4

1

2

Lógicamente, el ordinal asociado a cada valor entero es él mismo.

3.1.2.  El tipo de dato "Carácter" (CHAR)

Es el tipo utilizado para manejar caracteres sueltos, tales como letras, números del 0 al 9, signos de puntuación, etc..

El identificador predefinido correspondiente a este tipo de dato es CHAR y el ordinal aso-ciado a cada uno de los caracteres es su código ASCII, el cual establece el orden en el conjunto de valores CHAR. Así,  por ejemplo, se cumple que:

'A' < 'S'

'Z' < 'a'

ya que los códigos ASCII de 'A' y de 'Z' son respectivamente menores que los de 'S' y 'a', como puede verse en la siguiente tabla:


Téngase en cuenta que aunque el Pascal no distingue mayúsculas y minúsculas para iden-tificadores y palabras reservadas, dentro del tipo CHAR  sí son diferentes los caracteres 'A' y ''a'. Un dato de tipo CHAR ocupa en memoria 1 byte.

3.1.3.  El tipo de dato "Booleano" (BOOLEAN)
Los valores del tipo Booleano son únicamente dos: Verdadero y Falso, que se representan mediante los identificadores predefinidos (constantes predefinidas) TRUE y FALSE. El identifi-cador predefinido correspondiente a este tipo es BOOLEAN.

Los ordinales asociados a cada valor son:

Ordinal de FALSE
:  0

Ordinal de TRUE  
:  1

Por tanto, FALSE < TRUE. Un dato de tipo BOOLEAN ocupa en memoria 1 byte.

3.2.  DECLARACIONES ELEMENTALES
Como ya adelantamos en el tema anterior, en la zona de declaraciones de un programa Pascal, se declaran y/o definen los elementos de programación disponibles que van a ser utiliza-dos en la zona de instrucciones del programa. Estos elementos son:


                  
1.  Etiquetas             
4.  Variables


2.  Constantes            5.  Procedimientos

                  
3.  Tipos de dato     
6.  Funciones

El Pascal Estándar exige que la descripción de estos elementos se realice en este orden, mientras que Turbo Pascal permite hacerla en cualquier orden.

Veremos a lo largo del curso cómo se realiza la descripción de todos estos elementos. De momento, diremos que la descripción de cada uno de ellos va precedida de una palabra reser-vada:



Etiquetas  .......... 
LABEL         
Variables  ............ 
VAR

Constantes  .......
CONST         
Procedimientos ... 
PROCEDURE

Tipos de dato .... 
TYPE          
Funciones  ........... 
FUNCTION
Comenzaremos en este tema estudiando la declaración de variables y la definición de constantes.

3.2.1.  Declaración de Variables
Una variable en programación es un espacio de memoria reservado para contener datos de un cierto tipo, cuyo contenido puede variar durante la ejecución del programa al que pertenece y que es reconocida en el mismo por el identificador que el programador le ha asignado.

En Pascal, toda variable que vaya a ser utilizada en la zona de instrucciones del programa debe ser declarada previamente, lo cual se realiza escribiendo un identificador válido para la va-riable seguido de dos puntos (:), el tipo de variable y punto y coma (;).

Para indicar al compilador que lo que se está declarando son variables y no otra cosa, la declaración va precedida de la palabra reservada VAR.

Por ejemplo, si en un programa se va a utilizar una variable de tipo BOOLEAN, una de ti-po CHAR y dos de tipo INTEGER, la declaración de variables será:

                           PROGRAM COBI;

                            VAR         B 
:  BOOLEAN;

                                             C 
:  CHAR;

                                      NUM 
:  INTEGER;

                                         RES 
:  INTEGER;

                           BEGIN
                            < Zona de Instrucciones >

                           END.

donde COBI, B, C, NUM y RES son identificadores válidos elegidos por el programador.

Todas las variables del mismo tipo se pueden agrupar, separando sus identificadores con comas:

                            VAR   NUM,RES 
:  INTEGER;

                                                      B 
:  BOOLEAN;

                                                      C 
:  CHAR:

Durante el proceso de compilación se reservará el espacio de memoria correspondiente a cada variable, según su tipo, que a partir de este momento será reconocido por su correspondien-te identificador. El contenido de estas variables no es significativo (es aleatorio) hasta que sean cargadas en la zona de instrucciones. Así, para las variables B y C del ejemplo anterior se reser-va 1 byte y para NUM y RES 2 bytes. Hasta que alguna instrucción les proporcione un valor, se puede considerar que están vacías.

3.2.2.  Definición de Constantes
Una constante en programación es un espacio de memoria reservado para contener un da-to concreto de un cierto tipo, cuyo contenido es fijado de antemano y no puede variar durante la ejecución del programa al que pertenece, y es reconocida por el mismo por el identificador que el programador le ha asignado.

Definir constantes puede ser útil cuando un cierto dato, quizá pesado de escribir, va a ser usado repetidas veces en un programa. Si le asociamos un identificador sencillo, el programa ga-nará en claridad y ahorraremos tiempo de escritura.

Para definir constantes en Pascal se escribirá, después de la palabra reservada CONST, el identificador de la constante seguido del signo igual (=), del valor de la constante y punto y co-ma (;).

Así, en un programa podríamos definir la constante:

PROGRAM CURRO;

                               CONST  EXPO = 2545693;

                               VAR             CT : INTEGER;

                           BEGIN
                            < Zona de Instrucciones>

                           END.

Cuando se define una constante, no hay que indicar el tipo de la misma, ya que con el va-lor de la constante el tipo es reconocido por el compilador. La variable definida en el ejemplo se-ría del tipo LONGINT.

Podemos definir constantes de cualquier tipo simple y también constantes alfanuméricas (cadenas de caracteres), las cuales deben ir siempre entre apóstrofes:

PROGRAM ZD;

              CONST  REGLA 
=  'Esta Casa Tiene Varias Puertas Falsas';

                                  CAD 
=  '203';

                                 NUM 
=  203;

            < Resto del Programa >

Nótese la diferencia entre las constantes CAD y NUM. La primera es una constante alfanu-mérica y NUM es de tipo BYTE. Con NUM podemos realizar las operaciones disponibles para los tipos enteros, mientras que con CAD no.

Debe observarse también la diferencia que existe entre la descripción de variables y cons-tantes: las variables se declaran y las constantes se definen.

A la hora de compilar la zona de definicion de constantes, no sólo se reserva el espacio de memoria correspondiente a las constantes, sino que además se cargan con el valor definido.

3.3.  INSTRUCCIONES ELEMENTALES
Veremos en esta sección tres instrucciones básicas del lenguaje Pascal: la Asignación, la Escritura en Pantalla y la Lectura de Variables por Teclado.

3.3.1.  Instrucción de Asignación

Esta instrucción permite almacenar datos en variables previamente declaradas. El operador que realiza la asignación consiste en dos puntos, seguidos inmediatamente por el signo igual (:=). La sintáxis de esta instrucción es:


Identificador  de Variable := Dato;

donde el Dato debe ser del mismo tipo que la Variable donde se almacena. Por ejemplo, en el programa COBI, podríamos hacer las siguientes asignaciones:

                           PROGRAM COBI;

                             VAR  NUM,RES: INTEGER;

                                                      B : BOOLEAN;

                                                      C : CHAR;

                           BEGIN
                             NUM := -128;

                             B := FALSE;

                             RES := 32165;

                             C := 'T';

                           END.

En este punto hemos de decir dos cosas: primera, que las variables de tipo BOOLEAN, co-mo en este caso la variable B, sólo podrán ser cargadas por asignación; segunda, que cuando manejemos caracteres (y también cadenas de caracteres) en un programa, deben ir entre apóstro-fes, como en este caso el carácter 'T'.

3.3.2.  Escritura en Pantalla
El lenguaje Pascal escribe en pantalla mediante los procedimientos predefinidos WRITE y WRITELN. La sintáxis de estos procedimientos es:


 WRITE ( Item, Item, ... );

WRITELN ( Item, Item, ... );

donde Item es algo que se quiere escribir en pantalla: una expresión literal, el valor de una varia-ble en un momento dado, etc..

Todos los Items que se escriban dentro del paréntesis de un WRITE o WRITELN separa-dos por comas, se escribirán en pantalla uno a continuación de otro, en la misma línea y sin dejar espacios en blanco entre ellos. Por ejemplo:

WRITE ('Hola Mundo',1,2,3);

producir  la salida por pantalla siguiente:

Hola Mundo123

Si queremos dejar espacios en blanco entre items debemos reflejarlo expresamente. Una manera es:

WRITELN (Item,' ',Item,' ', ... );

Por ejemplo:

WRITE ('Hola Mundo, ',1,' ',2,' ',3);

producirá la salida por pantalla:

Hola Mundo, 1 2 3

La única diferencia existente entre WRITE y WRITELN es que con WRITE el cursor per-manece en la línea en la que se han escrito los ítems, justo a continuación del último carácter es-crito mientras que con WRITELN, después de escribir los ítems correspondientes, se produce un salto de línea.

Otra característica de las sentencias WRITE y WRITELN es que para cada item podemos especificar la anchura (en espacios) que queremos que ocupe en la pantalla, con la siguiente sin-táxis:


WRITELN ( Item:anchura, Item:anchura, ... );
donde anchura es una expresión entera (literal, constante, variable, etc.).

Ejemplo:  El programa:

                         PROGRAM COOB;

                           CONST M = 100;

                           VAR A,B,C : BYTE;

                         BEGIN
                           A:=10; B:=2; C:=32;

                           WRITE (A:3, B:5, C:10, M:10);

                         END.

producirá la siguiente salida por pantalla:

                                            10              2                      32                    100

              

                                           3            5                 10                      10               espacios

Si la anchura de un ítem es mayor que la anchura especificada, el Pascal expande la anchu-ra al tamaño mínimo necesario. Si en el ejemplo anterior hubiésemos puesto  WRITE (A:3, B:5, C:10, M:2); habríamos obtenido:

10    2        32100
La instrucción WRITELN; sin datos, deja una línea en blanco: significa algo como escribe nada y salta a la línea siguiente.

3.3.3.  Lectura por Teclado

Igual que con la escritura por pantalla, para la lectura de variables por teclado existen dos procedimientos predefinidos: READ y READLN. Utilizaremos a lo largo del curso el procedi-miento READLN con la sintáxis:

READLN (Variable);

donde Variable se carga cuando pulsemos Return después de teclear el dato que debe almace-narse en la variable. Por ejemplo:

                   PROGRAM LUGO;

                      VAR C:CHAR;

                                 I:INTEGER;

                    BEGIN
                      WRITE('Pulsa un carácter: ');

                      READLN(C);

    WRITE('Introduce un número entero: ');

                      READLN(I);

                      WRITE(C,' ',I);

                      READLN;

                    END.

Si hemos tecleado el carácter 'K' y el número 123, en pantalla tendremos:

Pulsa un carácter: K

Introduce un número entero: 123

K 123

¿Cuál será el resultado de la instrucción READLN; ?

3.4.  EXPRESIONES
Una expresión es un conjunto de operandos conectados entre sí por un conjunto de ope-radores, que da como resultado un valor del mismo o distinto tipo que los operandos, dependien-do de la naturaleza de los operadores.

La mayoría de los operadores en Pascal son binarios, es decir, conectan a dos operandos; el resto son unitarios (sólo afectan a un operando).  Los operadores disponibles en Pascal pueden clasificarse de la siguiente manera:


Aritméticos

Lógicos

Operadores            Relacionales

Para el tipo STRING

De dirección


Veremos en este capítulo los operadores disponibles para los tipos de dato que de momen-to conocemos.

3.4.1.  Operadores Aritméticos para los Tipos Enteros
Los operadores aritméticos disponibles para los tipos numéricos enteros son:      


+     
:   SUMA

 -     
:   RESTA

*    
:   MULTIPLICACION

DIV   
:   DIVISION ENTERA (COCIENTE)

MOD
:   RESTO DE LA DIVISION ENTERA

Todos estos operadores son binarios y si los dos operandos son de distintos tipos enteros, el resultado es del tipo entero común. Así, el tipo común de BYTE e INTEGER es INTEGER y de INTEGER y WORD es LONGINT.

Existe también el operador unitario de cambio de signo (-).

Ejemplos:         7 + 11    =  18                     7 - 11   =  -4

                          7 * 11    =  77                 7 DIV 11 =   0

                      7 MOD 11 =   7                         - (7)  =  -7

Si en una expresión aparecen varios operadores aritméticos, para ejecutarlos se establece el siguiente orden de prioridad:

*, DIV y MOD  :  Mayor prioridad

                                                          +  y  -          :  Menor prioridad

Aunque más prioritario que los operadores binarios es el operador unitario de cambio de signo. Si aparecen seguidos operadores de igual prioridad, se ejecutan de izquierda a derecha.

Ejemplo:  El resultado de la siguiente expresión aritmética es -11.

2 + 6 - 5 * 4 + 7 MOD 2  =  2 + 6 - 20 + 1  =  -11

Sin embargo se puede alterar el orden de cálculo, encerrando entre paréntesis lo que se de-sea que se calcule en primer lugar; es decir, las expresiones entre paréntesis se evalúan las pri-meras para convertirlas en un operando simple.

Ejemplo:  Poniendo paréntesis en la expresión anterior, obtendremos otro resultado:

2 + ( 6 - 5 ) * ( 4 + 7 MOD 2 )  =  2 + 1 * 5  =  7

Como consecuencia de las definiciones y reglas anteriores, es evidente la igualdad:

A  =  ( A DIV B ) * B + ( A MOD B )

En efecto, desde la enseñanza primaria sabemos que Dividendo es igual a Divisor por Co-ciente más Resto. Nótese que los paréntesis no son necesarios.

3.4.2.  Operadores Lógicos o Booleanos
Los operadores booleanos conectan operandos de tipo BOOLEAN y el resultado de sus ex-presiones es también BOOLEAN. Los operadores de este tipo disponibles en Turbo Pascal son:


NOT    :   NEGACION LOGICA

AND    :   AND LOGICO

OR       :   OR LOGICO

XOR    :   XOR LOGICO

Los operadores AND, OR y XOR son binarios y NOT unitario. La operación  B AND C  da resultado TRUE sí y sólo sí ambos valores son TRUE simultáneamente;  B OR C  da resultado TRUE si al menos uno de los valores es TRUE;  B XOR C  da resultado TRUE si uno de los dos valores es TRUE, pero no si ambos son TRUE. El operador unitario NOT aplicado a un valor boolean devuelve TRUE si el valor era FALSE y viceversa.

En los operadores booleanos, al igual que en los aritméticos, también se establece un orden de prioridad. El más prioritario es NOT, luego AND y, por último OR y XOR (igual prioridad). Igualmente, si en una expresión aparecen operadores de igual prioridad, se evaluarán de izquier-da a derecha. La prioridad de los operadores booleanos se puede modificar utilizando parén-tesis.

Ejemplo:  Suponiendo que  A,B y D  tienen el valor FALSE y que C tiene el valor TRUE, eva-luar la siguiente expresión:

A OR B XOR C AND NOT D AND A

Primero se evaluaría NOT :

A OR B XOR C AND TRUE AND A

Luego AND de izquierda a derecha:

A OR B XOR TRUE AND A

A OR B XOR FALSE

Luego OR y XOR de izquierda a derecha:

FALSE XOR FALSE

Es decir:                                          FALSE

Ejemplo:  Si en la expresión anterior utilizamos paréntesis, podemos cambiar el resultado:

A OR ( B XOR C ) AND NOT ( D AND A )

Evaluando primero los paréntesis:

A OR TRUE AND NOT FALSE

Luego NOT :

A OR TRUE AND TRUE

Después AND:                           A OR TRUE

Y, por último OR:                            TRUE

3.4.3.  Operadores Relacionales o de Comparación

Los operadores relacionales son los operadores de comparación. Son operadores binarios que se pueden aplicar a operandos de cualquier tipo ordinal, y dan como resultado un valor BOOLEAN. Estos operadores son:


>     
:
MAYOR

>=    
:   
MAYOR O IGUAL

<     
:   
MENOR

<=    
:   
MENOR O IGUAL

=     
:   
IGUAL

<>    
:   
NO IGUAL

Ejemplo:  Programa que reciba dos datos enteros y escriba por pantalla TRUE si uno de los dos es múltiplo del otro y FALSE en caso contrario.

                 PROGRAM MULTIPLO;

                   VAR A,B:INTEGER;

                 BEGIN
                   READLN(A); READLN(B);

                   WRITE (((A MOD B) = 0) OR ((B MOD A) = 0));

                 END.

Como vemos, A MOD B es un número entero y el 0 es un valor entero. La expresión (A MOD B) = 0 da resultado TRUE si A es múltiplo de B y FALSE en caso contrario (resultado BOOLEAN). Análogamente ocurre con (B MOD A)=0. En la misma instrucción, por último, te-nemos una expresión booleana que da resultado BOOLEAN.

3.5.  TIPOS SIMPLES DE DATOS DEFINIDOS POR EL PROGRAMADOR
Como ya adelantamos en la sección 3.1., en PASCAL, además de los tipos simples prede-finidos, es posible crear tipos ordinales nuevos, que pueden ser bien subconjuntos de los tipos  predefinidos, o bien tipos de dato exclusivos, sin ninguna relación con los tipos predefinidos. A los primeros se les denomina tipos Subrango y a los segundos tipos Abstractos o Enumerativos.                            

Comenzaremos estudiando estos últimos.

3.5.1.  Tipos Abstractos o Enumerativos
Supongamos que tenemos que escribir un programa para obtener resultados estadísticos sobre población, por provincias, a partir de datos recogidos en Galicia. Es muy probable que ne-cesitemos disponer de una variable capaz de contener cuatro valores que asociaríamos directa-mente a las cuatro provincias gallegas; por ejemplo, para poder tomar la decisión de imprimir los resultados de Lugo, o de visualizar por pantalla los resultados de Orense.

Con los tipos de datos predefinidos, podríamos declarar la variable PROV, de tipo CHAR y asociar a las cuatro provincias los caracteres 'C', 'L', 'O' y 'P'. De esta forma, podríamos tener:

Leer del teclado el Código de Provincia (PROV)

Si PROV='C' entonces imprimir resultados de Coruña.

Si PROV='L' entonces imprimir resultados de Lugo.

Con esta solución evidente, tendríamos por una parte que recordar en cada momento el có-digo asociado a cada provincia y, por otra, que controlar los errores que se producirían si al car-gar la variable PROV se introduce un carácter diferente a 'C', 'L', 'O' o 'P'.

En PASCAL podemos obtener otra solución, quizá  más elegante, definiendo un nuevo ti-po de dato, que se ajuste a nuestras necesidades. Podemos crear el tipo de dato PROGAL cuyos posibles valores serían CORUÑA, LUGO, ORENSE y PONTEVEDRA. ¿ Cómo se hace esto ?

La definición de tipos de dato abstractos se realiza, en la zona de definiciones y declaracio-nes del programa, escribiendo en primer lugar la palabra reservada TYPE, a continuación el identificador del tipo, después un signo igual (=) y después, los identificadores de los valores po-sibles separados entre sí por comas y encerrados entre paréntesis. Siguiendo con el ejemplo, definiríamos el tipo así:

                PROGRAM GALICIA;

                   TYPE PROGAL = (CORUÑA,LUGO,ORENSE,PONTEVEDRA);

                   VAR PROV : PROGAL;

                < Resto del Programa >

Como vemos, la definición de un tipo abstracto se hace por enumeración de todos sus po-sibles valores. Esta es la razón por la que también se les denomina tipos enumerativos.

Veamos alguna característica más de estos tipos de datos:

i)  Los tipos abstractos son simples y ordinales: el compilador asociar  los ordinales 0, 1, 2, ...  a los valores del tipo en el orden en que han sido enumerados. Así, en el ejemplo:

Ordinal de Coruña = 0      Ordinal de Orense        = 2

Ordinal de Lugo     = 1      Ordinal de Pontevedra = 3

Por esta razón, los operadores relacionales vistos anteriormente se pueden utilizar con operan-dos abstractos.

ii)  Los datos de un tipo abstracto ocupan en memoria 1 byte si el tipo tiene enumerados hasta 256 valores, y 2 bytes si tiene más de esta cantidad, hasta 65.536, que es el número máximo posible.

iii)  Debemos tener en cuenta que los posibles valores de un tipo enumerativo deben ser exclusi-vos de ese tipo. Así, no sería posible crear el tipo VOCALES formado por los valores 'A', 'E', 'I', 'O' y 'U' ya que estos lo son del tipo CHAR.

iv)  Las variables de un tipo abstracto se pueden cargar por asignación; por ejemplo:

PROV := ORENSE;

pero no se pueden leer con READLN ni escribir con WRITE y WRITELN directamente. Puede servir como razón que READLN, WRITE y WRITELN son procedimientos predefini-dos de lectura y escritura y servirán, por tanto, para leer y escribir datos de tipos predefini-dos. Para leer y escribir datos de tipos abstractos tendremos que diseñar procedimientos es-pecíficos. Esto lo veremos posteriormente, en los temas dedicados a "Procedimientos y Fun-ciones".

3.5.2.  Tipos Subrango
Supongamos que queremos realizar un programa para llevar la contabilidad en un edificio de 15 pisos en el que existen cinco viviendas por planta. Nos interesaría manejar una variable PISO, de tipo entero, cuyos valores estuviesen comprendidos entre 1 y 15, y una variable LE-TRA, de tipo carácter, cuyos valores estuviesen comprendidos entre 'A' y 'E', y que no pudieran contener valores fuera de esos límites. Tales variables se dice que son de tipos subrango.

En general un tipo subrango es un subconjunto de valores de un tipo ordinal denominado tipo base (host type). En la definición de un tipo subrango se escriben, después del identificador del tipo y del signo igual (=), los valores mínimo y máximo que pertenecen al subrango, separa-dos por dos puntos seguidos. Ejemplos de tipo subrango serían:

TYPE   PLANTA = 1 .. 15;

VIVIENDA = 'A' .. 'E';

Notemos tres cosas de estos tipos:

i)  Al definir un tipo subrango no se especifica el tipo base, que es reconocido automáticamente mediante los valores mínimo y máximo.

ii)  Una variable de tipo subrango tiene todas las propiedades de las variables del tipo base, aun-que sus valores deben estar en el rango especificado. Así las variables:

VAR   PISO     : PLANTA;

 LETRA : VIVIENDA;

     tienen las propiedades de los tipos BYTE y CHAR respectivamente: espacio de memoria ocu-pado, operadores disponibles, etc..

iii) Una característica importante de los tipos subrango es que no es necesario definirlos como ti-pos y, por tanto, podemos declarar variables subrango directamente. Así, para el ejemplo an-terior, podríamos hacer directamente:

                            PROGRAM CONTABLE;

                               VAR  PISO  : 1..15;

                                      LETRA : 'A'..'E';

                            < Resto del Programa >

Obviamente, los tipos abstractos, al ser ordinales, admiten tipos subrango. Veamos un ejemplo:

PROGRAM NAME;

   TYPE  DIASEM =  (LUNES,MARTES,MIERCOLES,JUEVES,VIERNES,SABADO,DOMINGO);

   VAR    DIALAB  :  LUNES .. VIERNES;

< Resto del Programa >

DIALAB es una variable de tipo subrango del tipo base DIASEM.

3.6.  TIPOS REALES
El PASCAL define el tipo de dato REAL (identificador predefinido) para manejar núme-ros reales: aquellos que tienen una parte entera y una parte decimal. Los números reales se deno-minan frecuentemente números de coma flotante.

Estos números son manejados internamente en notación científica, en la cual un numero, N, se representa de la forma:

N = M x BE
donde M se denomina Mantisa y su valor puede estar comprendido entre:

1 <= M < B

B es la Base del sistema de numeración utilizado y E es el Exponente. Por ejemplo, en el sistema decimal, el número 459.27 se representa como:

459.27 = 4.5927 x 102
En el sistema binario, el número 1001 se representa como:

1001 = 1.001 x 23
El numero de dígitos de la mantisa (cifras significativas) y el rango de valores del expo-nente dependerán del compilador utilizado. Turbo Pascal utiliza 6 bytes para almacenar datos de tipo REAL: 5 para la mantisa y 1 para el exponente.   

       bit de signo                       mantisa  (5 bytes)                         exponente (1 byte)                                    


Expresado un número real N de esta forma, su valor viene dado por la fórmula:


Si  0 < E <= 255, entonces N = (-1)s x 2(E-129) x (1.M)

Si  E = 0, entonces N = 0

Con este formato obtenemos para los datos REAL el rango de valores y cifras significati-vas siguientes:

TIPO
RANGO
CIFRAS SIGNIF.
TAMAÑO (BYTES)

REAL
2.9 x 10-39 ... 1.7 x 1038
11
6

En efecto, el valor mínimo positivo distinto de cero que puede alcanzar N según la fórmula anterior es cuando E tome el valor 1, M el valor 0 y s el valor 0: en este caso N llegará a ser:

N = (-1)0 x 2(1-129) x (1.0) = 2-128 = 2.9 x 10-39
Igualmente el valor máximo positivo que puede alcanzar N es cuando E tome el valor 255, M el valor 99999999999, de forma que 1.M = 1.99999999999 ( 2, y s el valor 0; es decir:

N = (-1)0 x 2(255-129) x (1.99999999999) = 2126 x 2 = 1.7 x 1038
Nótese que este rango de valores sólo están reflejados los números reales positivos, desde el más pequeño (2.9 x 10-39) al más grande posible (1.7 x 1038), y que corresponde al bit de signo 0. También existe el mismo rango para números negativos, que corresponde al bit de signo 1. Las 11 cifras significativas corresponden a que con 39 bits se pueden representar números deci-males hasta el 239 que es un número de 11 cifras.

3.6.1.  Constantes y Variables de tipo REAL
Como resulta lógico, las constantes de tipo REAL deben ser diferenciadas de las de cual-quier tipo entero, para que el compilador las almacene correctamente en la memoria. Para ello, las constantes REAL tendrán punto decimal, exponente o ambos.

      Veamos ejemplos válidos de definiciones de constantes de tipo REAL:

CONST  ALFA    
=  33.0;

BETA    
=  6E+02;

GAMMA  
=  -3.940392E+18;

DELTA   
=  3.;

EPSILON 
=  .5E-3;

ZETA    
=  -.3e-1;

THETA     
=  6.e-20;

Si para ALFA y DELTA hubiésemos puesto:

CONST    ALFA     = 33;

DELTA  =  3;

serían constantes enteras que ocuparían en memoria 1 byte.

Como vemos, la flexibilidad para escribir números reales es grande: solamente se exige no dejar espacios antes y después de la "E" (o "e") que denota al exponente.

Las variables de tipo REAL se pueden cargar por asignación o por teclado con cualquiera de los formatos vistos en el ejemplo anterior. Además, a una variable de tipo REAL le podemos asignar o cargar por teclado un dato de cualquier tipo entero: la conversión a REAL se realiza automáticamente.

En el siguiente programa-ejemplo la constante entera 31 se carga en la variable real X y el resultado de la expresión entera (20 DIV 7) se almacena en la variable real Y.

           PROGRAM JAEN;

               VAR  S,T,U,V,W,X,Y,Z : REAL;

           BEGIN
WRITELN ('Carga de variables reales con distintas notaciones.');

S:=33.0;

T:=6E+02;

U:=3.;

V:=-.3e-1;

WRITE ('Dame un número real en cualquier notación válida: ');

READLN (W);

X:=31;

Y:=20 DIV 7;

WRITE ('Dame un número entero: ');

READLN (Z);

           END.

3.6.2.  Escritura en Pantalla del tipo REAL
Como sabemos desde la sección 3.3., los procedimientos de escritura en pantalla en len-guaje PASCAL son WRITE y WRITELN. Utilizados con valores REAL directamente, obtene-mos la salida en forma exponencial, con 11 cifras significativas. Por ejemplo, utilizando las va-riables del programa anterior, las instrucciones:

WRITELN (S);  WRITELN (V);   WRITELN (Y);

nos darían en pantalla los valores:

3.3000000000E+01

-3.0000000000E-02

2.0000000000E+00

Las instrucciones WRITE y WRITELN con el formato:

WRITE (Item:anchura);

nos devuelven también notación exponencial. Así:

WRITELN (S:8);  WRITELN (V:8);  WRITELN (Y:10);

escribirían en pantalla:

3.3E+01

-3.0E-02

2.000E+00

Nótese que en esta notación para los números reales, el signo de la mantisa ocupa siempre un carácter, aunque en los números positivos no se escriba el signo (+).

Con este formato, la anchura se expandirá al tamaño mínimo necesario (8 espacios) cuan-do ésta se especifique incorrectamente (anchura<8). Por ejemplo:

WRITE (V:4)

escribirá:

-3.0E+01

Para obtener en pantalla números REAL en notación decimal (sin exponente) existe un ter-cer formato, en el que se especifica la anchura y el número de cifras decimales:      

   

WRITE ( R : Anchura : Decimales );

donde R es cualquier dato de tipo REAL. Por ejemplo:

WRITE (S:5:2, S:6:2, V:10:3);

nos daría:

                                                            33.00   33.00      -0.030

                                                                5         6           10         espacios

Nótese que con esta notación, el signo (+) de número positivo no ocupa espacio, aunque el (-) de número negativo si. Si la anchura se especifica incorrectamente, se expande a la mínima necesaria; es decir,

WRITELN (S:3:3);

nos da:

-0.030

con anchura 6.

3.6.3.  Operadores para el tipo REAL

Los operadores aritméticos disponibles para el tipo REAL son:


+  :
SUMA

- 
:  
RESTA

* 
:   
MULTIPLICACION

/  
:   
DIVISION

Si el operador de división (/) se utiliza con operandos enteros, el resultado es un valor de tipo REAL.

Los operadores multiplicativos (* y /) son igual de prioritarios, y son más prioritarios que los aditivos (+ y -). Existe también el operador unitario de cambio de signo, que es más prio-ritario que todos los operadores binarios. El resto de las reglas de prioridad (ejecución de iz-quierda a derecha, uso de paréntesis, etc.) son las mismas que para los tipos enteros.

Los operadores relacionales o de comparación pueden utilizarse con valores de tipo REAL, para dar un resultado BOOLEAN.

3.6.4.  Otros tipos Reales
Aunque de momento no los vamos a utilizar (lo posponemos para un tema posterior), dire-mos que Turbo Pascal soporta otros cuatro tipos reales, cuyas características vemos en la si-guiente tabla:

TIPO
RANGO
CIF. SIG.
TAMAÑO (bytes)

  SINGLE

  DOUBLE

  EXTENDED

  COMP
1.5 x 10-45 ... 3.4 x 1038
5.0 x 10-324 ... 1.7 x 10308
3.4 x 10-4932 ... 1.1 x 104932
- 263 + 1 ... 263 + 1 
7

15

19

19
4

8

10

8

La utilización natural de estos tipos de dato implica la existencia en nuestro PC de copro-cesador matemático 80X87, además de instruir al compilador para que genere el código máquina utilizable por el 80X87. Esto se hace activando una opción de compilación denominada Proceso Numérico; opción que se encuentra desactivada por defecto.

Incluso en ausencia de 80X87 podremos utilizar los cinco tipos reales, instruyendo al com-pilador para que emule por software al coprocesador, activando las opciones de compilación Emulación 80X87 y Proceso Numérico. En este estado, las velocidades de compilación y ejecu-ción disminuyen sensiblemente, por lo que no debe activarse a menos que sea absolutamente ne-cesario.

3.7.  COMENTARIOS
Un comentario en programación es una anotación que se puede escribir en cualquier parte del programa para recordarnos qué significan ciertas variables, qué hace cierto conjunto de ins-trucciones, etc., y que no tienen ningún efecto sobre la ejecución del programa.

En Turbo Pascal podemos comenzar un comentario con el símbolo de apertura de llave     ( { ), el cual informa al compilador que ignore todo lo que encuentre a continuación hasta el sím-bolo de cierre de llave ( } ).

Existe también una forma alternativa: abrir comentario con el símbolo doble (* y cerrarlo con *).

Ejemplo:                PROGRAM MULTIPLO;

  VAR A,B:INTEGER;                { Declaración de Variables }

BEGIN
  READLN(A); READLN(B);                (* Lectura de datos *)

  WRITE (((A MOD B) = 0) OR ((B MOD A) = 0));

 END.

3.8.  VARIABLES INICIALIZADAS O CONSTANTES CON TIPO
Si al definir una constante lo hacemos con la siguiente sintáxis:


CONST   Identificador : Tipo = Valor;

lo que estamos haciendo es declarar una variable del tipo especificado, inicializada con el valor especificado. En la zona de instrucciones esta constante con tipo se comporta exactamente igual que una variable, pudiendo ser la receptora en una instrucción de asignación. Téngase en cuenta que una tal variable sólo se inicializa una vez: en la zona de declaraciones del programa.

Ejemplo:                             PROGRAM SORIA;

  CONST  MAXI : BYTE = 255;

                 DECI   : REAL = -8.6;

                 CARA : CHAR = '$';

BEGIN
  WRITELN (MAXI);         { En pantalla 255  }

  WRITELN (DECI:4:1);    { En pantalla -8.6  }

  WRITELN (CARA);         { En pantalla $      }

  MAXI:=100;                       (*   Cambio   *)

  DECI:=69;                           (*      de        *)

  CARA:='V';                         (*    Valor     *)

  WRITELN (MAXI);          { En pantalla 100  }

  WRITELN (DECI:2:1);     { En pantalla 69    }

  WRITELN (CARA);          { En pantalla V     }

END.

APENDICE 3.1.


LA UNIDAD SYSTEM   ( I )

Veremos en este apéndice los elementos de la programación incluidos en la Unidad SYS-TEM, que de momento podemos empezar a utilizar. En posteriores apéndices iremos ampliando esta Unidad. La Unidad System contiene todos los elementos estructurales del Lenguaje Pascal y cualquier programa la puede utilizar directamente, sin necesidad de declararlo mediante la claú-sula USES.

Debemos hacer ahora una primera distinción entre Funciones y Procedimientos. Una Fun-ción es un subprograma que entrega al programa principal que la utiliza un valor y sólo uno de un cierto tipo; puede recibir o no parámetros del programa principal, y debe ir asignada a una va-riable del programa, del mismo tipo que el resultado, o bien debe ser enviado su resultado a pan-talla con WRITE o WRITELN; es decir:


Variable := IDENTIFICADOR_FUNCION (Parámetros);

WRITE (IDENTIFICADOR_FUNCION (Parámetros));

Un Procedimiento es un subprograma que realiza un cierto cometido; puede entregar o no valores al programa principal que lo utiliza; puede recibir o no parámetros y su utilización en el programa depende de la naturaleza del procedimiento. Frecuentemente se utiliza de la forma:


IDENTIFICADOR_PROCEDIMIENTO;

IDENTIFICADOR_PROCEDIMIENTO (Parámetros);

A.3.1.1.  Constantes Predefinidas
IDENTIFICADOR
TIPO
VALOR

MAXINT

MAXLONGINT

PI
INTEGER

LONGINT

REAL
32.767

2.147.483.647

3,1415926536

MAXINT contiene el mayor valor entero positivo posible para el tipo INTEGER;  MAX-LONGINT el mayor valor entero positivo posible para el tipo LONGINT y PI una aproximación del número (, utilizado frecuentemente en geometría.
Ejemplo de utilización:   WRITE (MAXINT);

C:=2*PI*RADIO;          { C y RADIO variables REAL }

A.3.1.2  Variables Inicializadas
IDENTIFICADOR
TIPO
VALOR INICIAL

RANDSEED
LONGINT
0

RANDSEED es la semilla (seed) de la que parte la función generadora, RANDOM, de nú-meros aleatorios. Si ejecutamos la función RANDOM dos o más veces consecutivas sin cambiar RANDSEED, obtendremos los mismos números aleatorios. 

Ver ejemplo de utilización en A.3.1.8.

A.3.1.3.  Funciones Aritméticas
Todas las funciones que aquí aparecen reciben un parámetro, que denotaremos X, y de-vuelven el resultado de las funciones matemáticas más usuales.

IDENTIFICADOR
PARAMETRO
RESULTADO
SIGNIFICADO

ABS

ARCTAN

COS

EXP

FRAC

INT

LN

SIN

SQR

SQRT
Real o Entero

Real o Entero

Real o Entero

Real o Entero

Real o Entero

Real o Entero

Real o Entero

Real o Entero

Real o Entero

Real o Entero
Real o Entero

Real

Real

Real

Real

Real

Real

Real

Real o Entero

Real
Valor Absoluto

Arcotangente

Coseno

Exponencial

Parte Decimal

Parte Entera

Logaritmo Neperiano

Seno

Cuadrado

Raíz Cuadrada

ABS (X), donde X puede ser Real o Entero, nos devuelve el valor absoluto de X, en el mismo ti-po de dato que el parámetro; por ejemplo:

ABS (-6.99) = 6.99      
 ABS (9.9) = 9.9

ABS (-2723) = 2723       
ABS (203) = 203

ARCTAN (X), nos devuelve el ángulo expresado en radianes (recuérdese que  360° = 2(  radia-nes) cuya tangente trigonométrica es X. El resultado es siempre Real. Por ejem-plo:

ARCTAN (1)   = PI/4                                   { 45° }

ARCTAN (2.0) = 1.1071487178E+00    { 63°26'5.8" }

COS (X), nos da el coseno de X (ángulo que debe venir expresado en radianes). El resultado es de tipo Real. Por ejemplo:

COS (0) = 1.0       COS (PI/2) = 0.0       COS (PI) = -1.0

EXP (X), devuelve la exponencial de X; es decir, el valor Real e elevado a X, donde e es la base de los logaritmos neperianos o naturales (e = 2.7182818284). Así:   

EXP (0) = 1.0       EXP (1) = e

FRAC (X),  nos da la parte fraccional de X,  es decir:   FRAC (X) = X - INT (X):

FRAC (-8) = 0.0       FRAC (2.000) = 0.000       FRAC (-123.456) = -0.456

INT (X), nos da la parte entera de X. El resultado es Real. Por ejemplo:

INT (-8) = -8.0       INT (2.000) = 2.000       INT (-123.456) = -123.000

LN (X), nos devuelve en tipo Real, el logaritmo neperiano de X. Por ejemplo:

LN (1) = 0.0       LN (1.0) = 0.0

SIN (X), nos da el seno de X (ángulo que debe venir expresado en radianes). El resultado es de tipo Real. Por ejemplo:

SIN (0) = 0.0       SIN (PI/2) = 1.0       SIN (PI) = 0.0

SQR (X), nos devuelve el cuadrado de X, es decir, X*X. El resultado es del mismo tipo que el argumento; es decir si X es entero, SQR(X) es entero; si X es real, SQR(X) es real. Por ejemplo:

SQR (-2.0) = 4.0       SQR (69) = 4761

SQRT (X), retorna la raíz cuadrada de X. El resultado es Real. Por ejemplo:

SQRT (169.0) = 13.0       SQRT (169) = 13.0

A.3.1.4.  Funciones de Transferencia

Se denominan Funciones de Transferencia aquellas que aplicadas a un parámetro de un cierto tipo, devuelven un valor de un tipo diferente. Todas las que hay en la Unidad System son:

IDENTIFICADOR
PARAMETRO
RESULTADO
SIGNIFICADO

CHR

ORD

ROUND

TRUNC
BYTE

Tipo Ordinal

REAL

REAL
CHAR

LONGINT

LONGINT

LONGINT
Carácter Asociado

Ordinal Asociado

Redondear Parámetro

Truncar Parámetro

CHR (N),  donde N es un dato de tipo BYTE, retorna el carácter cuyo valor ASCII asociado es N. Por ejemplo:

CHR (27 + 9) = $       CHR (64) = @       CHR (165) = Ñ

ORD (O), donde O es de cualquier tipo ordinal, nos da el número ordinal asociado al valor ence-rrado entre paréntesis. Por ejemplo, si tenemos definido el tipo 

GALICIA = (CORUÑA, LUGO, ORENSE, PONTEVEDRA)

en un programa, tendremos:

ORD (LUGO) = 1       ORD (PONTEVEDRA) = 3

ORD ('@') = 64           ORD (CHR(123))   = 123

ROUND (R), donde R es un dato de tipo REAL, retorna un valor de tipo LONGINT, que es el valor de R redondeado al número entero más cercano. Si R está exactamente en la mitad de dos números enteros, el resultado es el número de mayor valor absoluto. Si el redondeo de R no está dentro del rango LONGINT se producirá un error de ejecución. Por ejemplo:

ROUND (27.76) = 28       ROUND (1.627328E+5) = 162733

ROUND (-7.55) = -8         ROUND (7.55) = 8

TRUNC (R),  donde R es un valor de tipo REAL, devuelve un valor de tipo LONGINT que es el valor de R truncado por el punto decimal. Si el trunque de R no está dentro del rango LONGINT, se producirá un error de ejecución. Por ejemplo:

TRUNC (27.76) = 27        TRUNC (1.627328E+5) = 162732

TRUNC (-7.55) = -7          TRUNC (1.999999E12) = Error

A.3.1.5.  Procedimientos Ordinales

Estos procedimientos modifican el valor de una variable de cualquier tipo ordinal. Los dos que existen en la Unidad System son:

IDENTIFICADOR
SIGNIFICADO

DEC

INC
Decrementa una variable

Incrementa una variable

DEC (O,E),  donde O es una variable de cualquier tipo ordinal y E es cualquier valor de tipo en-tero, produce el decremento de O en E unidades; es decir, DEC(O,E) corresponde a O:=O-E. Si el valor E no se especifica el decremento por defecto es en una unidad; i.e., DEC(O) correspondería a  O:=O-1;

Ejemplo de utilización : En el siguiente programa se decrementan con DEC varia-bles de los tipos INTEGER, CHAR y BOOLEAN:

PROGRAM LEON;

VAR I:INTEGER; C:CHAR; B:BOOLEAN;

BEGIN
WRITELN('El programa carga las variables.');

I:=100; C:='L'; B:=TRUE;

WRITELN('I = ',I); WRITELN('C = ',C); WRITELN('B = ',B);

WRITELN('Modificamos las variables con DEC');

DEC(I);   WRITE('I = ',I);

DEC(C,4); WRITE('C = ',C);

DEC(B);   WRITE('B = ',B);

END.

Obtendremos en pantalla después de la decrementación:

I = 99       C = H       B = FALSE

INC (O,E), donde O es una variable de cualquier tipo ordinal y E es cualquier valor de tipo ente-ro, produce el incremento de O en E unidades; es decir, INC(O,E) corresponde a O:=O+E. Si el valor E no se especifica el incremento por defecto es en una unidad; i.e., INC(O) correspondería a  O:=O+1;

Ejemplo de utilización :

PROGRAM ZAMORA;

VAR I:INTEGER; C:CHAR; B:BOOLEAN;

BEGIN
WRITELN('El programa carga las variables.');

I:=100; C:='L'; B:=FALSE;

WRITELN('I = ',I); WRITELN('C = ',C); WRITELN('B = ',B);

WRITELN('Modificamos las variables con INC');

INC(I);   WRITE('I = ',I);

INC(C,4); WRITE('C = ',C);

INC(B);   WRITE('B = ',B);

END.

Obtendremos en pantalla después de la incrementación:

I = 101      C = P       B = TRUE

A.3.1.6.  Funciones Ordinales
Estas funciones se aplican a un parámetro de algún tipo ordinal. Devuelven al programa principal un valor del mismo o distinto tipo. Son las siguientes:

IDENTIFICADOR
PARAMETRO
RESULTADO
SIGNIFICADO

ODD

PRED

SUCC
Tipo Entero

Tipo Ordinal

Tipo Ordinal
BOOLEAN

Mismo Tipo

Mismo Tipo
Prueba de Paridad

Valor Precedente

Valor Sucesor

ODD (E), donde E es un dato de cualquier tipo entero, nos devuelve el valor booleano TRUE si E es un número impar (odd) y FALSE en caso contrario. Ejemplo:

ODD (MAXINT) = FALSE       ODD(-MAXINT+1) = TRUE

PRED (O), donde O es un valor de cualquier tipo ordinal, devuelve un valor del mismo tipo or-dinal: el predecesor de O. Por ejemplo, definido el tipo GALICIA:

PRED (23) = 22                    
PRED ('a') = 'Z'

PRED (TRUE) = FALSE        
PRED (LUGO) = CORUÑA

SUCC (O), donde O es un valor de cualquier tipo ordinal, devuelve un valor del mismo tipo or-dinal: el sucesor (valor siguiente) de O. Por ejemplo:

SUCC (23) = 24         SUCC ('a') = 'b'

A.3.1.7.  Procedimientos Varios

Veremos un solo procedimiento, que no recibe parámetros del programa que lo utiliza:

IDENTIFICADOR
SIGNIFICADO

RANDOMIZE
Carga la Variable RANDSEED con un valor aleatorio

RANDOMIZE, sitúa en la semilla (variable RANDSEED) de la que parte el generador de números aleatorios, RANDOM, un valor aleatorio que obtiene a partir del reloj del sistema. Su ejecución es equivalente a RANDSEED:=Número aleatorio. 

Ver ejemplo de utilización en la sección A.3.1.8.

A.3.1.8.  Funciones Varias

IDENTIFICADOR
PARAMETRO
RESULTADO
SIGNIFICADO

RANDOM

SIZEOF

UPCASE
Ninguno o WORD

Variable o Tipo

CHAR
REAL o WORD

WORD

CHAR
Generador Aleatorio

Tamaño del Parámetro

Cambio a Mayúsculas

RANDOM o RANDOM (W), es una función que puede recibir o no parámetro. Si no re-cibe argumento, el resultado es un número REAL aleatorio dentro del rango:

0 <= X < 1

Si recibe argumento, W, que debe ser un número entero positivo, el resultado es un núme-ro aleatorio de tipo WORD dentro del rango:

0 <= X < W

Si el valor de W es igual a 0, el resultado de la función es 0.

Ejemplo:       
PROGRAM SALAMANCA;

BEGIN
WRITELN('Primero "aleatorizo" la semilla');

RANDOMIZE;

WRITE('Ahora obtengo un número real aleatorio: ');

WRITELN(RANDOM);

WRITELN('Ahora cargo la semilla con un valor concreto');

RANDSEED:=12;

WRITE('Por último obtengo un número entero aleatorio: ');

WRITELN(RANDOM(100));

END.

SIZEOF (Variable o Tipo), admite como parámetro el identificador de cualquier variable o de cualquier tipo de dato.  El resultado, de tipo WORD, es el número de bytes de memoria que ocupa el parámetro. Por ejemplo, si hemos declarado las variables A de tipo BYTE y B de tipo BOOLEAN así como el tipo de dato abstracto GALICIA, podríamos obtener que:

SIZEOF (INTEGER) = 2      SIZEOF (A) = 1       SIZEOF (REAL)       = 6

SIZEOF (GALICIA)  = 1      SIZEOF (B) = 1       SIZEOF (LONGINT) = 4

UPCASE (C), donde C es un dato de tipo carácter, retorna otro dato de tipo carácter: el ar-gumento en mayúsculas. Si el parámetro no se encuentra en el rango 'a'..'z', no se verá afectado por la función. Por ejemplo:

UPCASE ('r') = 'R'       UPCASE ('T') = 'T'        UPCASE(':') = ':'

EJERCICIOS

1.- ¿Qué diferencia hay entre una constante predefinida y un identificador predefinido ?

2.- ¿ Por qué no existen para el tipo REAL los operadores MOD y DIV ?

3.- ¿ Cuál es el resultado de las siguientes operaciones ?:

       a)  3 + 5 * 5 / 2                           b)  7 - 1000 * 1000 / 10000

       c)  ROUND (2580.98)               d)  TRUNC (4030700001.125)

       e)  10000 * 10000.0 / 10000       f)  3 MOD 4 + (5.0 * 2 / 4) DIV 3

4.-  FRAC y INT  ¿qué son ?: constantes predefinidas o palabras reservadas.

5.-  En Pascal no existe la función elevar un número a una cierta potencia: A elevado a B. Con las funciones predefinidas disponibles, ¿cómo podríamos simular esta función ?

6.-  Como vimos en el tema 3, los operadores de comparación también se pueden utilizar con el tipo de dato REAL, para dar un resultado BOOLEAN. ¿Cuál sería el resultado de la expre-sión siguiente? :

100.0 / 3.0 * 3.0 = 100.0

7.-   Programa que calcule el área lateral, el área total y el volumen de un cilindro.

8.-   Programa que convierta un tiempo, dado en segundos, al formato: Horas:Minutos:Segundos.

9.-   Programa que calcule el área y el volumen de una esfera.

10.- Programa que refleje los máximos valores enteros con que trabaja su versión de TURBO PASCAL

11.- Programa que transforme una cantidad, introducida en pesetas, a dólares (1 dolar = 115 pts), francos (1 franco = 19.5 pts), libras (1 libra = 208 pts), marcos (1 marco = 66 pts), liras (100 liras = 8.8 pts), escudos (100 escudos = 79.4) y yens (100 yens = 93.6 pts).  (Cambio del 12.11.88)

12.- Programa que transforme una velocidad, expresada en m/sec, en Km/h.

13.- Programa que presente en pantalla la estructura general de todo programa PASCAL.

14.- Programa que refleje en pantalla el conjunto de palabras reservadas del Turbo PASCAL 7.0.

